Surface Area of a Solid of Revolution

Objective

This lab presents a second maplet for visualizing solids of revolution with three-dimensional plots. The other elements of this maplet display the definite integral for its surface area and its value — exact and approximate.

Background

Consider the solid formed when a smooth curve $y = f(x)$, $a \le x \le b$, is revolved about the x-axis. The (lateral) surface area of this solid is given by the definite integral

$$S = 2\pi \int_a^b f(x)\sqrt{1 + (f'(x))^2}\, dx.$$

The `SurfaceOfRevolution` maplet is a convenient way to visualize and compute the volume of a solid of revolution about either the x- or y-axis. This is similar to the `VolumeOfRevolution` maplet in that the function entered in the maplet must represent the distance from the curve to the axis. the axis, i.e., the (outer) radius of the solid. The `SurfaceOfRevolution` command in the `Student[Calculus1]` package does allow for the specification of inner and outer radii; this command can produce a plot, definite integral, or surface area.

Discussion

Enter, and execute, the following Maple commands in a Maple worksheet.

Example 1: The `SurfaceOfRevolution` maplet

In this example the solid produced when the graph of $y = x^2 + 1$ on $[0, 3]$ is revolved around the x-axis is displayed and its surface area computed.

- From your browser, launch the `SurfaceOfRevolution` maplet.
- In the **Function** field, enter `x^2+1`.
- Set $\mathbf{a} = 0$ and $\mathbf{b} = 2$.
- Check that the **Horizontal axis** box is checked.
- To see the solid, click the **Plot** button.
- To see the definite integral and its value (exact and approximate), click the **Area** button.

Example 2: Horizontal Axis of Revolution

```
> restart;                                      # clear Maple's memory
> with( Student[Calculus1] );                   # load package
> R2 := x -> x^2 + 1;                           # define outer radius
> SurfaceOfRevolution( R2(x),                   # 3-D plot (rotate it!)
>         x=0..2, output=plot );
> q1 := SurfaceOfRevolution( R2(x),             # surface area as integral
>        x=0..2, output=integral );
> q2 := value( q1 );                            # exact area
> evalf( q2 );                                  # floating-point approximation
```

Example 3: Sphere (General Radius)

```
> R3 := x -> sqrt( r^2 - x^2 );                 # upper semi-circle
> q3 := SurfaceOfRevolution( R3(x),             # surf area as integral
>        x=-r..r, output=integral );
> simplify( q3 );                               # simplify integrand
> q4 := simplify( q3, symbolic );               # simplify integrand!
> q5 := value( q4 );                            # exact area
```

Example 4: Torus ($a = 1$, $b = 2$)

```
> with( plots );                                      # load package
> top := x -> sqrt( 1 - x^2 );                        # upper semi-circle
> bot := x -> -sqrt( 1 - x^2 );                       # lower semi-circle
> P1 := SurfaceOfRevolution( top(x)+2,                # outer half of torus
>             x=-1..1, output=plot ):
> P2 := SurfaceOfRevolution( bot(x)+2,                # inner half of torus
>             x=-1..1, output=plot ):
> display( [ P1, P2 ], scaling=constrained,           # display torus
>  title="Torus as Surface of Revolution" );          # display torus
> q6 := SurfaceOfRevolution( top(x)+2,                # surf area as integral
>             x=-1..1, output=integral );
> q7 := SurfaceOfRevolution( bot(x)+2,                # surf area as integral
>             x=-1..1, output=integral );
> q8 := q6 + q7;                                      # total surface area
> q9 := combine( q80 );                               # 2 integrals into 1
> q10 := simplify( q9 );                              # simplify integrand
> q11 := value( q10 );                                # exact surface area
```

Notes

(1) The `SurfaceOfRevolution` maplet complains if the problem contains parameters. Fortunately, the `SurfaceOfRevolution` command is capable of working with problems with parameters (except that a plot cannot be created).

(2) In normal operation, Maple assumes all variables are complex-valued. A true appreciation of this is beyond the scope of this course. Instead of trying to provide all of the assumptions necessary to convince Maple to make desired simplifications, the `symbolic` option can be used to tell Maple to simplify an expression without regard to general restrictions that apply in the complex-valued case.

(3) The online help for the `SurfaceOfRevolution` commands reminds us that the definite integrals for surface areas usually can not be evaluated exactly in terms of elementary functions. When the integral can be evaluated, the result is often expressed in terms of "special functions" that are beyond the scope of this course. This means that `evalf` will need to be used to obtain a meaningful result in most surface area problems.

(4) If `evalf` is applied to a definite integral, such as the ones returned by `SurfaceOfRevolution` with `output=integral`, then the integral is evaluated using a numerical method. The result of this computation can be slightly different from the floating-point approximation to the exact value of the integral (as determined by applying `value` to the definite integral and then `evalf` to this result).

Questions

(1) Find a definite integral for the lateral surface area of a right circular cylinder with radius r and height h. What is the value of this integral? What is the total surface area of the cone?

(2) The lateral surface of a cone with radius r and height h can be unrolled into a sector of a circle. What is the radius of this circle? Express the fraction of the full circle as a function of r and h. Call this function $F(r, h)$. What are $\lim_{h\to 0^+} F(r, h)$ and $\lim_{r\to 0^+} F(r, h)$? What relationship between r and h ensures that the lateral surface is exactly 75% of the full circle?

(3) A *torus* (doughnut) can be obtained by revolving a circle with radius a about the line $x = -b$, with $0 < a < b$. Find the definite integral for the surface area of a torus and its value as a function of a and b. (What is the volume if $a = 1$ and $b = 2$?)